

JBoss Drools Expert



ВИКТОР ПОЛИЩУК

Виктор Полищук



- 10+ лет коммерческой разработки
- 8+ лет работы с Java
- 3+ лет программирования на Drools
- Участвовал/видел 50+ «проблемных» (из-за высокой нагруженности логикой) проектов
- 0 из 9 «провальных» проектов на Drools
- Технический лидер в Infopulse, Ukraine
- Контрактор в BICS, Belgium

JokerMobile™



«Простой»

2.00
за SMS

«Хитрый»

0.00
за первые **10** SMS/день

1.50
за SMS внутри сети

2.50
за SMS

Модель Данных: Customer



```
public class Customer {  
    String number;  
    Tariff tariff;  
}
```

Модель Данных: SMS



```
public class Sms {  
    String sender;  
    String receiver;  
    DateTime sent;  
}
```

Модель Данных: BillingRecord



```
public class BillingRecord {  
    Customer customer;  
    BigDecimal price;  
    Sms sms;  
}
```

Java FTW



- Пишем God класс
- Делим логику на стратегии по тарифам
- Делим логику на атомарные стратегии
- Что-то еще

API



```
public interface TariffStrategy {  
    Collection<BillingRecord> calculate(  
        Customer customer,  
        Interval interval  
    );  
}
```


«Хитрая» Стратегия: стр. 1



```
Collection<Sms> messages =  
    dao.findBySender(customer.getNumber(), interval);  
Collection<SmsBillingRecord> records =  
    new ArrayList<SmsBillingRecord>();  
Map<DateTime, List<Sms>> dailyMap =  
    new HashMap<DateTime, List<Sms>>();  
for (Sms sms : messages) {  
    DateTime date = sms.getSent().withMillisOfDay(0);  
    List<Sms> collection = dailyMap.get(date);  
    if (collection == null) {  
        collection = new ArrayList<Sms>();  
        dailyMap.put(date, collection);  
    }  
    collection.add(sms);  
}
```

«Хитрая» Стратегия: стр. 2




```
for (List<Sms> value : dailyMap.values()) {
    Collections.sort(value, SMSComparator.INSTANCE);
    for (int i = 0; i < value.size(); i++) {
        Sms sms = value.get(i);
        if (i < 10) {
            records.add(new BillingRecord(customer,
                BigDecimal.ZERO, sms));
        } else if (dao.findByNumber(sms.getReceiver()) != null) {
            records.add(new BillingRecord(customer,
                new BigDecimal("1.5"), sms));
        } else {
            records.add(new BillingRecord(customer,
                new BigDecimal("2.5"), sms));
        }
    }
}
return records;
```

Полный Успех



ПРАЗДНИК



Бонус



- Может быть применен к любому тарифу
- Могут складываться друг с другом
- Правила наложения могут быть уникальны

Модель Данных: Customer+



```
public class Customer {  
    String number;  
    Tariff tariff;  
    Collection<Bonus> bonuses;  
}
```

Бонус: «Двадцать»



- Платеж **100.00** в месяц
- Скидка **20%** на все SMS

Модель Данных: BillingRecord+



```
public abstract class BillingRecord {
    Customer customer;
    BigDecimal price;
}
public class SmsBillingRecord
    extends BillingRecord {
    Sms sms;
}
public class FixedBillingRecord
    extends BillingRecord {
}
```

Java FTW



- Растим God класс
- Пишем все в стратегии
- Выносим цены в некий PriceManager
- Создаем слой модифицирующий BillingRecord
- Что-то еще

Частичный Успех



БЕСПОКОЙСТВО



Бонус: «Два по Пятьдесят»



После порога в **50 SMS** в день:

- Удваиваем количество бесплатных SMS
- Скидка **50%** на последующие SMS до конца дня

Java WTF



- Убиваемся о God класс
- Меняем PriceManager
- Добавляем LimitManager
- Меняем «модифицирующий» слой:
 - Передаем доп.информацию про стратегии и бонусы
 - Усложняем логику
- Начинаем сначала
- Что-то еще

Я Сделал Все Что Мог



ДЕПРЕССИЯ... НЕНАВИСТЬ... РАЗДРАЖЕНИЕ...
МИЗАНТРОПИЯ... ЗАМКНУТОСТЬ...
УСТАЛОСТЬ... АПАТИЯ...




Drools Expert



- Продукционная экспертная система
- Построена на базе «Rete-сети»
- Отличная документация
- Открытый код
- Бесплатная
- Поддерживается JBoss

Глоссарий



Базовый Синтаксис



rule “<Имя правила>”

<опции>

when

<условие или LHS>

then

<следствие или RHS>

end

Drools: Тариф «Простой»



```
rule "Billing: charge -Simple-"
when
    $customer : Customer(tariff==Tariff.SIMPLE)
    Charge($sms : sms, customer==$customer)
then
    Log.info("Charge {} as {}", $sms, 2.0);
    insertLogical(
        new SmsBillingRecord($customer, 2.0, $sms)
    );
end
```


Charge?



```
public class Charge {  
    Customer customer;  
    Sms sms;  
}
```

<ИЛИ>

```
declare Charge  
    customer : Customer @key  
    sms : Sms @key  
end
```

Drools: А Кто Платить Будет!?!...



```
rule "Default: sender is in charge"  
when  
    $period : Interval()  
    $sms : Sms($sender : sender, $period.contains(sent))  
    $customer : Customer(number==$sender)  
    not Charge(customer!=$customer, sms==$sms)  
  
then  
    Log.info("Mark {} as regular", $sms);  
    insertLogical(new Charge($customer, $sms));  
end
```

Drools: Тариф «Хитрый» стр. 1



```
rule "Tariff -Smart-: free SMS"
when
    Day($period : interval)
    $customer : Customer(tariff==Tariff.SMART)
    Charge($sms : sms, customer==$customer,
           $period.contains(sms.sent))
    accumulate($c : Charge(customer==$customer,
                           $period.contains(sms.sent),
                           !sms.sent.isAfter($sms.sent));
               $count : count( $c ); $count<=10)
then
    Log.info("Mark {} as free", $sms);
    insertLogical(new ChargeAsFree($customer, $sms));
end
```

Drools : Тариф «Хитрый» стр. 2



```
rule "Tariff -Smart-: local SMS"
```

```
when
```

```
Day($period : interval)
```

```
$customer : Customer(tariff==Tariff.SMART)
```

```
Charge($sms : sms, customer==$customer,  
        $period.contains(sms.sent))
```

```
not ChargeAsFree(customer==$customer, sms==$sms)
```

```
exists Customer(number==$sms.receiver)
```

```
then
```

```
Log.info("Mark {} as local", $sms);
```

```
insertLogical(new ChargeAsLocal($customer, $sms));
```

```
end
```

Drools: Тариф «Хитрый» стр. 3



```
rule "Billing: free of charge"  
when  
    $customer : Customer()  
    ChargeAsFree($sms : sms, $customer==customer)  
  
then  
    Log.info("Charge {} as {}", $sms, 0.0);  
    insertLogical(  
        new SmsBillingRecord($customer, 0.0, $sms)  
    );  
end
```

Drools: Тариф «Хитрый» стр. 4



```
rule "Billing: charge local in -Smart-"
when
    $customer : Customer(tariff==Tariff.SMART)
    ChargeAsLocal($sms : sms, customer==$customer)
    not ChargeAsFree(sms==$sms)

then
    Log.info("Charge {} as {}", $sms, 1.5);
    insertLogical(
        new SmsBillingRecord($customer, 1.5, $sms)
    );
end
```

Drools: Тариф «Хитрый» стр. 5



```
rule "Billing: charge others in -Smart-"
when
    $customer : Customer(tariff==Tariff.SMART)
    Charge($sms : sms, customer==$customer)
    not ChargeAsFree(sms==$sms)
    not ChargeAsLocal(sms==$sms)

then
    Log.info("Charge {} as {}", $sms, 2.5);
    insertLogical(
        new SmsBillingRecord($customer, 2.5, $sms)
    );
end
```

Drools: Бонус «Двадцать»



```
rule "Subscription -20%-: refund"  
when  
    $customer : Customer(bonuses contains Bonus.TWENTY)  
    Charge($sms : sms, customer==$customer)  
    not ChargeAsFree(sms==$sms)  
    not Modifier(sms==$sms, modifier<0.8)  
  
then  
    Log.info("Add modifier {} to {}", 0.8, $sms);  
    insertLogical(new Modifier($sms, 0.8));  
end
```


Drools: Тариф «Хитрый» стр. 4+




```
rule "Billing: charge local in -Smart-"
when
    $customer : Customer(tariff==Tariff.SMART)
    ChargeAsLocal($sms : sms, customer==$customer)
    not ChargeAsFree(sms==$sms)
    Modifier($modifier : modifier, sms==$sms)
then
    BigDecimal price = $modifier * 1.5;
    Log.info("Charge {} as {}", $sms, price);
    insertLogical(
        new SmsBillingRecord($customer, price, $sms)
    );
end
```

Domain Specific Language



ВЫ ЧТО, ВСЕ ЕЩЕ КОДИТЕ БИЗНЕС-ПРАВИЛА?



Drools: Тариф «Хитрый» стр. 4+



```
rule "Billing: charge local in -Smart-"
when
    $customer : Customer(tariff==Tariff.SMART)
    ChargeAsLocal($sms : sms, customer==$customer)
    not ChargeAsFree(customer==$customer, sms==$sms)
    Modifier($modifier : modifier, sms==$sms)

then
    BigDecimal price = $modifier * 1.5;
    Log.info("Charge {} as {}", $sms, price);
    insertLogical(
        new SmsBillingRecord($customer, price, $sms)
    );
end
```

DSL: Тариф «Хитрый» стр. 4+



```
rule "Billing: charge local in -Smart-"
```

```
when
```

```
  Find a customer
```

```
    - with Tariff.SMART
```

```
  Find customer's local SMS
```

```
  Proceed only if it is not free
```

```
  Find an SMS modifier
```

```
then
```

```
  Create SMS billing record with base price: 1.5
```

```
end
```

DSL Словарь



```
[when] Find a customer = $customer : Customer()
[when] - with {tariff} = tariff=={tariff}
[when] Find customer's local SMS = ChargeAsLocal($sms :
  sms, customer==$customer)
...
...
...
...
...
[then] Create SMS billing record with base price: {base}
= BigDecimal price = $modifier * {base};
Log.info("Charge {} as {}", $sms, price);
insertLogical(new SmsBillingRecord($customer, price,
  $sms));
```

Тестирование



ЛЕКАРСТВО ОТ РАБСТВА



Подсказки



- Работаем с **production knowledge base**
- Готовим **крошечные** тестовые данные
- Выбираем **дружелюбные** критерии прохождения
- Тестируем **business case**, а не Drools
- Performance тесты **тоже** важны

Тестируем Drools:



```
private final Customer callee = new Customer("007", Tariff.SIMPLE);
private final Interval billingPeriod = new Interval(
    new DateTime(2013, 9, 1, 0, 0, 0, 0),
    new DateTime(2013, 10, 1, 0, 0, 0, 0)
);
@Test
public void testSimple() {
    DateTime instant = new DateTime(2013, 9, 20, 4, 40, 11);
    BigDecimal expectedResult = new BigDecimal("2.0");
    BigDecimal actualResult = executeGetAmount(callee,
        generateSMS("007", "any", instant, 0, 1),
        billingPeriod,
        callee
    );

    assertEquals(expectedResult, actualResult);
}
```


Итоги



ЕЩЕ ЧУТЬ-ЧУТЬ И ПЕРЕРЫВ



Достоинства



- Отделение логики от «бизнес-логики»
- Независимые правила
- Менять правила может даже аналитик
- Тесты свободные от Моск-ов
- «Теплый, ламповый» Java код
- Приемлемая производительность
- ... и другая куча «плюшек»

Смотри Также



- JBoss Drools Guvnor
- JBoss Drools Fusion
- OptaPlanner (JBoss Drools Planner)
- jBPM

- <http://www.jboss.org/drools/>
- <http://www.jcp.org/en/jsr/detail?id=94>
- <https://github.com/victor-cr/drools-joker-mobile>



Спасибо



ЗА ВНИМАНИЕ